

HOPE COMPUTER COMPANY LTD.

汉字2.13H源程序详解



鲍岳桥 编写
郑国荣

北京希望电脑公司

汉字 2.13H 源程序详解

鲍岳桥
郑国荣 编写

北京希望电脑公司

一九九二年三月

内 容 提 要

计算机在我国的普及应用，首先就要解决好汉字信息的计算机处理问题。2.13H是我国微机用户最常使用的普及型汉字操作系统，它集中体现了当前我国众多汉字操作系统的许多优点，并形成了自己的特色。了解和研究2.13H的内部细节，对每一个与汉字操作系统打交道的人来说都将大有裨益。

本书针对2.13H（CC版）在微机上实现的每一步骤进行了系统而全面的解释和分析，揭示了其内部的全部秘密，甚至还提到了若干设计错误。全书共分为七章，第一章介绍一些基础知识，第二～七章分别对显示字库读取、打印字库读取、键盘管理、显示、打印驱动、特殊显示、光标闪烁、文件打印等模块程序作了详尽解释。为了方便起见，程序解释中用到的一些常用资料和数据均可在本书有关章节内找到，使读者不必钻入堆积如山的资料之中就能通读本书。

本书的主要读者对象是从事与汉字有关的计算机科研、设计和应用的工作人员，尤其是计算机系统软件的设计人员，也可作为大专院校师生计算机课程的教学参考书。对于已初步了解计算机系统和初步掌握8086汇编语言的计算机学习者，本书也可起到进一步引导和深化的作用。

前 言

2.13 系列汉字操作系统自 1986 年问世以来,经过不断完善和发展,获得了很大的成功。由于它具有功能强大、使用简便、适应面广等特点,到目前为止,已成为拥有最广泛用户的汉字操作系统。然而,要想真正发挥 2.13 系列汉字操作系统的全部优越性能,做到左右逢源、运用自如,甚至根据用户自身的特点加以适当改造,就必须深入了解其内部的技术特点。为此,我们对该系统在 PC 机上实现的全部过程进行了分析和整理,与广大用户和读者共勉。

2.13H 是 2.13 系列汉字操作系统中的最高软件版。根据机器显示方式的不同分为两种子版本,即 CC 版和 GW 版。CC 版主要用于原装机(没有配汉卡的机器),GW 版主要用于国产机(带汉卡的机器)。本书是针对 2.13H 的 CC 版编写的。

本书提供了 2.13H(CC 版)中主要文件的汇编源程序清单,并把这些程序按其功能结构进行了划分。各程序中每个子程序前都给出了功能和输入输出参数的说明。对程序内的几乎每条指令都作了较详细的解析。为了便于读者阅读时查找,程序中所有标号均用加载后内存绝对地址的形式给出。由于许多驻留程序的变量地址直接放在 PSP 中,这些单元在程序中无法找到,所以在源程序清单前还列出了本程序所使用变量的地址及相应的功能说明。另外,为节省篇幅,对那些较大的数据区,只给出其地址,而省略其内容。

本书共分为七章,第一章简要地介绍了一些与 2.13H 密切相关的基础知识,提供了以后需要使用的一些常用表格。第二、三章分别介绍了各种驻留方式显示字库的读取程序及打印字库读取程序,读者可根据所使用的机器特点和需要选择阅读。第四、六章分别详细介绍了键盘管理模块、显示模块、打印驱动模块程序。第七章介绍了特殊显示、光标闪烁、文件打印及其它一些有用的程序,供读者参考。以上所有程序整理完成后经汇编连接均能正确运行,如有必要,读者可直接按书中绝对地址加以修改。

在本书编写过程中,曾得到陈红雨等同志的大力支持,在此真诚地向他们表示谢意。

由于时间较紧,加之作者水平有限,书中难免会有错漏之处,恳请广大计算机同行指正。

作 者

一九九二年元月 于杭州

目 录

第一章 基础知识	1
第一节 磁盘布局	1
第二节 程序段前缀和文件控制块	3
第三节 内存控制块	5
第四节 本书引用的中断调用	5
第二章 显示字库读取	9
第一节 读硬盘字库 FILE0A.COM	9
第二节 一级字库驻留内存 FILE1A.COM	16
第三节 全部字库驻留内存 FILE2.COM	26
第四节 读虚盘字库 FILE3.COM	28
第三章 打印字库读取	34
第一节 读 16 点阵字库 FILE16B.COM	34
第二节 读 24 点阵字库 FILE24A.COM	41
第四章 键盘管理模块	59
第一节 使用说明	59
第二节 CCCC.COM 程序解析	59
第五章 显示管理模块	123
第一节 显示模块程序简介	123
第二节 VGA 26 行显示 CV26.COM	123
第六章 打印驱动程序	171
第一节 打印机驱动 PRTA.COM	171
第二节 屏幕拷贝 SEGP.COM	205
第七章 其它	215
第一节 安装工作参数 HHDOS.COM	215
第二节 汉化 C 盘 DOS 系统 HHCDOS.COM	217
第三节 菜单选择 MENUHH.COM	220
第四节 特殊显示 INT10F.COM	220
第五节 光标闪烁 INT1C.COM	254
第六节 文件打印 LP.COM	256

第一章 基本知识

本章将简要介绍一些与汉字操作系统 2.13H 密切相关的基础知识, 尤其是由 MS-DOS 约定的磁盘和存储器的数据组织布局或格式。

第一节 磁盘布局

由于 2.13H 必须在带有硬盘的计算机上运行, 故本节仅限于针对硬盘介绍其布局。

PC 的硬盘体系结构允许用户将一个物理硬盘划分为四个分区, 安装多个操作系统共享资源, 同时还可选择启动时所需的操作系统。由于 16 位 DOS 管理硬盘的最大容量为 32 MB, 因此划分分区的方法就为 DOS 管理大容量提供了条件。当 PC 上电在硬盘上引导系统时它首先从系统中第一个硬盘的第一个物理扇区读入硬盘主引导记录, 并将控制转给该记录。

一、主引导扇区 (分区扇区)

主引导扇区就是硬盘的第一个物理扇区, 即 0 头 0 柱面 1 扇区, 它在磁盘 (或内存) 中的映象如下图所示:

(7C00:10)	主引导记录代码	
8AH	0--8AH 共139字节	
1BDH	(错误提示信息) 8BH--1BDH 共307字节	
1BEH	分 区 表	
1FDH	1BEH--1FDH 共64字节	
1FEH	55	主引导记录有效标志 2字节
1F7H	AA	

可见, 主引导扇区主要由两部分内容组成。

第一部分是主引导记录代码, 它负责检查所选择分区 (称为“活动分区”或“可自举分区”) 是否唯一存在, 若是则把相应操作系统引导记录 (即该分区上的引导记录) 装入内存, 并把控制转给它。否则给出出错提示信息。

第二部分是分区表, 它位于主引导扇区的后部。由于硬盘最多可能存在四个分区, 每个分区表项的长度为 16 字节, 这样, 四个分区表项共占 64 字节。

每个分区表项布局如下:

偏移量	长 度	名 字	内 容 说 明
0(0)	1 字节	分区状态	0 = 不活动分区 80 = 活动分区, 可引导的
1(1)	1 字节	起始磁头	本分区起始磁头号
2(2)	1 字	起始扇区和起始柱面	用位编码标志来存储柱面号和扇区号: 字节 n CC S S S S S S 字节 n+1 C C C C C C C C 字节 n 的最前两位在前, 字节 n+1 的八位在后, 组成共十位的柱面号 CYLINDER. 字节 n 的最后六位构成扇区号 SECTOR

4(4)	1 字节	分区类型	01=12位FAT的DOS 02=XENIX 04=16位FAT的DOS 06=保留DOS	64=NOVELL 75=PCIX DB=CP/M FF=BBT
5(5)	1 字节	终止磁头	本分区终止磁头号	
6(6)	1 字	终止扇区和终止柱面	同起始扇区和起始柱面	
8(8)	双字	起始绝对扇区	注意字节交换	
C(12)	双字	扇区数	注意字节交换	

二、分区引导扇区(BOOT)

分区引导记录位于分区第一逻辑扇区，其布局如下：

偏移量	长 度	内 容 说 明
0(0)	3字节	跳转到引导记录代码
3(3)	8字节	版本号。也是OEM(原厂委托制造)名字或标志
B(11)	字	每扇区字节数
D(13)	1字节	每簇扇区数(必须是 2 的幂)
E(14)	字	保留扇区数(用于根目录、FAT等等)
10(16)	1字节	文件定位表(FAT)的个数(拷贝)
11(17)	字	根目录中的最大目录项数
13(19)	字	总扇区数
15(21)	1字节	介质描述字节
16(22)	字	每个FAT占用的扇区数
18(24)	字	每道扇区数
1A(26)	字	磁头数
1C(28)	字	隐含扇区数
1E(30)	224字节	其它：包括程序、数据或保留未用区
FE(254)	2字节	引导扇区数有效标志：55 AA

三、文件定位表(FAT)

文件定位表(File Allocation Table)简称 FAT，它是记录分区中文件占用簇号链的一张表，其中的项用于存放簇号。12 位 FAT 每项占 1.5 字节，16 位 FAT 每项占 2 字节。FAT 表(一般有两个拷贝)位于紧接 BOOT 扇区之后的连续扇区中。

(一) 12 位 FAT

项	例 值	用 途	说 明
0	FF8	磁盘标识字节	1.第0,1两项保留给DOS 2.第0项 FF8 表示硬盘
1	FFF	填充符	
2	003	指向第 3 簇	磁盘簇的值: 000=可用簇 001-FEF=下一簇号 FF0-FF6=保留簇 FF7=坏簇 FF8-FFF=文件的最后簇
3	004	指向第 4 簇	
4	FFF	文件结束簇	
5	000	可用簇(空簇)	

(二) 16 位 FAT

项	例 值	用 途	说 明
0	FFF8	磁盘标识字节	1.第0,1两项保留给DOS 2.第0项 FFF8表示硬盘
1	FFFF	填充符	
2	0003	指向第 3 簇	1.磁盘簇的值: 0000=可用簇 0001-FFFF=下一簇号 FFF0-FFF6=保留簇 FFF7=坏簇 FFF8-FFFF=文件最后簇 2.FAT项是字节交换的
3	0004	指向第 4 簇	
4	FFFF	文件结束簇	
5	0000	可用簇(空簇)	

四、文件目录表(FDT)

文件目录表(File Directory Table)简称 FDT,它是记录分区中文件(或子目录)目录及其属性的一张表。根目录 FDT 位于紧接 FAT 表之后的盘簇中,子目录 FDT 位于其父目录指定的盘簇中。FDT 中的目录项每项 32 字节,顺序存放。其布局如下:

偏移量	长 度	描 述	格 式	说 明
0(0)	8字节	文件名	ASCII字符	文件名第一字节指出目录项状态: 00H=目录尚未使用过 05H=文件名的第一个字节实际为ESH ESH=文件被使用过但已被删除 2EH=是目录(若下一字节也为2EH,则簇域中包含父目录簇号)
8(8)	3字节	扩展名	ASCII字符	
B(11)	1字节	属性字节	位代码: 位0=只读 位4=子目录 位1=隐含 位5=修改位 位2=系统 位6=保留 位3=卷标 位6=保留	
C(12)	10字节	保留区		
16(22)	字	最后修改时间	日期/时间 格式	
18(24)	字	最后修改时间	日期/时间 格式	
1A(26)	字	起始簇号	二进制整数	
1C(28)	双字	文件字节长度	二进制整数双字	

第二节 程序段前缀和文件控制块

一、程序段前缀 (PSP)

当 DOS 运行一个程序时,首先为暂驻程序留出足够的空间,该空间称为程序段,用于存放从磁盘装入的程序。在 DOS 装载程序前,首先在程序段的前 100H 即 256 字节建立一个控制块,这个控制就是程序段前缀 (Program Segment Prefix)。其布局如下:

偏移量	长 度	通常用法	描 述	说 明
0(0)	字	CD20H	INT20H终止地址	

2(2)	字		内存分配块结束地址	是段地址
4(4)	字节	00H	保留	
5(5)	5字节		DOS系统功能远调用入口	06H:起后用作段大小
A(10)	双字		INT22H中断向量原内容	程序结束处理
E(14)	双字		INT23H中断向量原内容	Ctrl-Break 处理
12(18)	双字		INT24H中断向量原内容	严重错误处理
16(22)	字		父进程的PSP	是段地址
18(24)	20字节	FF=可用	文件DOS内部数码表	每个文件占一字节,位7=1表示不继承
2C(44)	字		环境块地址	是段地址
2E(46)	字节		保留	
32(50)	字	14H,00H	向量表大小	DOS允许更大的表
34(52)	双字	12H,00H	向量表地址	DOS3.3允许有向量表地址
38(56)	字节		保留	
50(80)	字	CD21H	INT21H DOS 调用	
52(82)	字节	CBH	远返回	
53(83)	9字节		保留	
5C(92)	36字节		缺省未打开文件控制块 1	
6C(108)	20字节		缺省未打开文件控制块 2	覆盖第一个FCB
80(128)	字节		命令行参数长度	也是缺省DTA的首址
81(129)	127字节		命令行参数	空格开始,回车结束

DOS设置程序段前缀的目的有三:一是用于存放DOS为了管理进程所需的数据;二是作为用户程序在运行时要求DOS提供信息的缓冲区;三是用作文件管理系统中输入/输出及交换文件数据的区域。

对用户来说,以下几个区域显得特别重要:

1. 内存顶部地址(即偏移 02H:内存分配块结束地址);
2. 段大小(即偏移 06H:此域包含了本程序段的最大可用字节数);
3. DOS内部数码表(018H-2BH:不同于文件句柄);
4. 环境块段地址(即偏移 2CH处);
5. 文件控制块(FCB);
6. 命令行/默认的磁盘传输区(DTA)

二、文件控制块(FCB)

文件控制块(File Control Block)是传统的DOS文件管理机制中,用户程序和操作系统之间对话的接口格式。程序欲进行文件操作前,初始化FCB,按其标准格式给出驱动器代码、文件名、扩展名,构成“未打开的FCB”,然后将其地址传送给DOS以打开或生成文件。如果文件被成功地打开或建立,DOS就根据磁盘当前文件登记项填写FCB的某些域,形成“打开的FCB”,供用户程序进一步使用。在FCB的保留区中,有时也存放某些其他信息,这些信息是用于操作系统自身目的的,且随不同的DOS版本而异,用户程序不能加以修改。

(一) 未打开的FCB

偏移量	长度	名字	内 容
0(0)	1字节	驱动器号	逻辑驱动器号:0=缺省,1=A,2=B,3=C等等
1(1)	8字节	文件名	ASCII码字符,不足时用空格填充

9(9)	3字节	扩展名	ASCII码字符,不足时用空格填充
C(12)	25字节	保留	置为 0

(二) 打开的 FCB

偏移量	长度	名字	内 容
0(0)	1字节	驱动器号	逻辑驱动器号:0=缺省,1=A,2=B,3=C等等
1(1)	8字节	文件名	ASCII码字符,不足时用空格填充
9(9)	3字节	扩展名	ASCII码字符,不足时用空格填充
C(12)	字	当前块号	二进制值表示当前记录块(文件打开时置为 0)
E(14)	字	记录大小	每个记录的字节数(缺省=128)
10(16)	双字	文件大小	二进制值表示文件大小(以字节计)
14(20)	字	文件日期	含有文件上次修改日期的紧缩字
16(22)	字	文件时间	含有文件上次修改时间的紧缩字
18(24)	8字节	保留	供 DOS 内部使用
20(32)	1字节	当前记录号	二进制值,表示当前记录(文件打开时为 0)
21(33)	双字	随机记录号	二进制值,表示下一个将要读写的随机块

第三节 内存控制块

内存控制块是 DOS 为了有效管理内存资源而设置的数据区, DOS 为每一个分配的内存区建立一个内存控制块, 长度为 16 字节(一个段落), 直接处于所控制的内存区前部。其布局如下:

偏移量	长度	名字	内 容
0	字节	位置	如不是最后一块为 4DH, 如是则为 5AH
1	字	进程ID	PSP 段地址
3	字	分配总数	分配的段数
5	11字节	保留	保留

只要程序对内存块有任何分配、修改、释放的请求, 或启动 EXEC 功能调用或结束一个程序, DOS 均检查内存控制链, 如果有一个块表现出已被破坏或控制链被破坏, DOS 就显示出错信息: Memory allocation error, 且系统死机。

第四节 本书引用的DOS和BIOS中断调用

在本书引用的DOS和BIOS中断调用中, INT 05H、INT 10H、INT 16H、INT 17H 等类型将在以后章节具体涉及的程序中详细说明, 在此不再赘述。为节省篇幅, 以下列举的中断类型及各子中断调用严格限于本书程序的使用范围。

一、INT 13H BIOS 磁盘 I/O

AH	功 能	输 入 参 数	输 出 参 数
	把在AL中说明的扇区数从在DL中指定	AH=02H AL=要读的扇区数	AH=00H 无错 =01-0FFH 错误代码

2	的驱动器读到ES:BX指定的缓冲区中。	CH=柱面号(作为低8位,0为基数) CL=柱面/扇区号,其中: 7-6位=柱面号(高2位) 5-0位=扇区数 DH=磁头号(0为基数) DL=驱动器号 =80H 硬盘1 =81H 硬盘2 ES:BX=缓冲区指针	=硬盘状态(40:74H) AL=所传送数据的扇区数 CF=0 无错 =1 有错
3	从由ES:BX定义的缓冲区把在AL中指家的扇区数写入由DL指定的驱动器中。	AH=03H AL=要写的扇区数 CH=柱面号(作为低8位,0为基数) CL=柱面/扇区号,其中: 7-6位=柱面号(高2位) 5-0位=扇区数 DH=磁头号(0为基数) DL=驱动器号 =80H 硬盘1 =81H 硬盘2 ES:BX=缓冲区指针	AH=00H 无错 =01-0FFH 错误代码 =硬盘状态(40:74H) AL=所传送数据的扇区数 CF=0 无错 =1 有错

二、INT 15H BIOS系统服务

AH	功 能	输 入 参 数	输 出 参 数
87	在任意地址空间(0-16MB)之间传送数据。	AH=87H CX=要移动的16位字的数据量(0-8000H个字,64K) ES:SI=指向由调用者分配的30H字节的表(描述符表GDT,见下表)。	AH=00H 移动完成 =01H RAM奇偶错误 =02H 其他例外中断错 CF=0 无错 =1 有错 ZF=0 移动未完成 =1 移动完成

全局描述符表(GDT)结构

局部描述符序号	偏 移	描 述 符 名	填 写
1	00(0)	空描述符	用户
2	08(8)	GDT位置	BIOS
3	10(16)	源数据块描述符(DS)	用户
4	18(24)	目标数据块描述符(ES)	用户
5	20(32)	本功能的代码段描述符(CS)	BIOS
6	28(40)	本功能使用的用户堆栈段描述符(SS)	BIOS

局部描述符表(LDT)的内容

位 长	内 容
16	段 长
16	段地址低位
8	段地址高位
8	访问权
16	保留

三、INT 18H ROM-BASIC解释程序

功 能	输 入 参 数	输 出 参 数
进入 ROM-BASIC解释器	/	进入ROM-BASIC的控制

四、INT 20H DOS终止程序

功 能	输 入 参 数	输 出 参 数
释放占用的内存并把结束地址、Ctrl-Break处理程序地址以及严重错误处理中断地址从PSP中得到恢复。	CS=当前程序的PSP段地址	控制转到结束地址

五、INT 21H DOS功能调用

AH	功 能	输 入 参 数	输 出 参 数
02	显示输出	DL=输出ASCII字符	
05	打印机输出 (DOS 自动检查打印机并给出适当信息)	DL=输出ASCII字符	
09	显示字符串	DS:DX=字符串首址 以'\$'为字符串结束标志	
0C	清除输入缓冲区并请求指定的输入功能	AL=输入功能号 =01H读键盘字节并回响 =06H直接控制台 I/O =07H直接控制台输入但不回响 =08H读键盘字符位不回显 =0AH键盘字符存入缓冲区	
11	查找第一个目录项	DS:DX=未打开的 FCB首址	AL=0 找到 AL=FF 未找到 DTA DS:DX=FCB 首址
1A	置 DTA 地址	DS:DX=DTA 地址	
25	设置中断向量	DS:DX=中断向量 AL=中断类型号	
30	取 DOS 版本号		AH=子版本号 AL=主版本号 BX和CX使用后被置为 0
31	程序结束并驻留	AL=返回码 DX=驻留区大小	AL=终止码
35	取中断向量	AL=中断类型	ES:BX=中断向量(CS:IP)
36	取空闲磁盘空间	DL=驱动器号 =0 缺省 =1 A 盘 =2 B 盘	成功:AX=每簇扇区数 BX=有效簇数 CX=每扇区字节数 DX=总簇数 失败:AX=0FFFFH
3D	打开文件	DS:DX=ASCHZ串首址 AL=0 读 =1 写 =2 读/写	成功:AX=文件句柄 失败:AX=错误码

3E	关闭文件	BX = 文件句柄	失败:AX = 错误码
3F	读文件或设备	DS:DX = 数据缓冲区地址 BX = 文件句柄 CX = 读取的字节数	成功:AX = 实际读入字节数 AX = 0 已到文件尾 出错:AX = 错误码 DS:DX = 信息块地址
40	写文件或设备	DS:DX = 数据缓冲区地址 BX = 文件句柄 CX = 写入的字节数	成功:AX = 实际写入的字节数 出错:AX = 错误码
42	移动文件指针	BX = 文件句柄 CX:DX = 位移量 AL = 移动方式(0,1,2)	成功:DX:AX = 新指针位置 失败:AX = 错误码
43	置/取文件属性	DS:DX = ASCII中地址 AL = 0 取文件属性 = 1 置文件属性 CX = 文件属性	成功:AX = 文件属性 CX = 新的文件属性 失败:AX = 错误码
4C	带返回码结束	AL = 返回码	
57	置/取文件日期和时间	BX = 文件句柄 AL = 0 读取 = 1 设置(DX:CX)	DX:CX = 日期和时间 失败:AX = 错误码

六、INT 25H DOS绝对磁盘读

功 能	输 入 参 数	输 出 参 数
读磁盘扇区，而与磁盘的文件结构无关。	AL = 驱动器号 = 0 A盘 = 1 B盘 = 2 C盘等等 CX = 所读连续逻辑扇区数 DX = 所读第1个逻辑扇区号 DS:BX = 磁盘传输区(DTA)地址	DS:BX = 指向传输区域的首址 注: 1.标志寄存器被置入栈中; 2.除段寄存器(CS,DS,ES,SS)以外的所有寄存器被破坏。

七、INT 27H DOS程序驻留退出

功 能	输 入 参 数	输 出 参 数
促使DOS终止当前的程序，驻留于内存，而不被以后装入的其他程序所覆盖，仅对.COM程序有效	CS = 程序的 PSP 首址 DX = 欲驻留部分程序的字节数加 1	

第二章 显示字库读取

第一节 读硬盘字库 FILE0A.COM

一、使用说明

FILE0A a b ←

其中：a表示常用字区大小，以50个汉字为单位。

b表示内部词组区大小，以k为单位。

例如：FILE0A 82表示保留400个常用字的缓冲区，内部词组区为2k字节。

当计算机的内存小于或等于640K，同时又需较大的内存自由空间时，可选择使用本程序。因使用本程序时显示字库不驻留内存，故可为用户保留较大的内存空间。但又由于本程序直接使用INT 13H从硬盘上读取汉字的点阵数据，磁盘操作过于频繁，既影响硬盘寿命，也使汉字显示速度受到很大的影响。

二、程序功能

本程序主要包括两个部分，即初始化代码和INT 7FH代码。

初始化代码主要完成以下工作：

- 1) 建立显示字库文件(HZK16)的扇区链表。
- 2) 初始化常用字区。
- 3) 初始化内部词组区。
- 4) 设置INT 7FH。
- 5) 驻留退出。

程序运行后，便可用INT 7FH读取汉字的点阵数据(16×16点阵)。

INT 7FH的运行过程如下：

- 1) 检索常用字区，若所需汉字在常用字区中，那么直接返回该汉字点阵数据的地址。
- 2) 否则，计算该汉字点阵数据在字库文件中的相对位移，查链表得到包含该汉字扇区的逻辑扇区号，换算为磁盘物理地址后，用INT 13H将该扇区读入内存。
- 3) 将该汉字及其点阵数据保存在常用字区中，使该汉字成为常用字。
- 4) 返回所需汉字点阵数据的地址。

三、变量名表

地址	长度	意义
0082h	1	命令行参数1, 指定常用字区大小
0083h	1	命令行参数2, 指定内部词组大小
00F0h	2	每簇扇区数
00F4h	2	磁盘数据区起始扇区的逻辑扇区号
00F6h	2	暂存连续簇数
01E0h	26h	字库文件(HZK16)的文件控制块(FCB)
0200h	2	内部词组区首址即常用字区尾址
0202h	2	内部词组区尾址
0204h	2	常用字区首址
0206h	2	常用字已使用区尾址
0208h	2	常用字区更新地址
020Ah	2	每扇区字节数
020Ch	2	每道扇区数
020Eh	2	磁头数
02C2h	1Eh	字库文件(HZK16)的扇区链表
02F0h	8	错误信息, 'ERROR', 7, 'S'

四、程序清单

;FILE0A.ASM

CODE

SEGMENT

ASSUME CS:CODE,DS:CODE

ORG 100H

L_0100:

;读入硬盘主引导扇区,目的是为了查找DOS分区的起始物理地址

```

MOV AX,201H           ;AH=2:调用读扇区功能,AL=1:读
                        ;入1个扇区
MOV BX,800H           ;读入数据存放地址
MOV CX,1
MOV DX,80H            ;物理地址为:C盘0道0面1扇区
INT 13H               ;读入主引导扇区

```

;查找DOS分区

```

MOV SI,OFFSET DS:[1BEH] ;分区表起始位移

```

L_0111:

```

CMP BYTE PTR [BX+SI+4],1
JE L_0128              ;是12 bits FAT 的DOS分区
CMP BYTE PTR [BX+SI+4],4
JE L_0128              ;是16 bits FAT 的DOS分区
CMP BYTE PTR [BX+SI+4],6
JE L_0128              ;是保留的DOS分区

```

当前分区项不是DOS分区项

```

ADD SI,10H             ;SI=下一分区项位移
JMP SHORT L_0111       ;检查下一分区项

```

L_0128:

;读入分区引导扇区(BOOT),目的是为了知道当前分区的基本信息

```

MOV DH,[BX+SI+1]       ;起始磁头号
MOV CX,[BX+SI+2]       ;起始扇区号和磁道号
MOV AX,201H            ;读入该分区的第一个扇区,即该
                        ;分区的引导扇区
INT 13H

```

;读入文件定位表(FAT)

```

MOV AL,[BX+16H]         ;一个FAT占用扇区数
INC AX                  ;加上BOOT扇区

```

;此时,CX,DX还是当前分区的起始物理地址,由于FAT正好紧接在BOOT扇区的后面,故这
;里为避免复杂的物理地址计算,将BOOT扇区和FAT一起读入内存.

```

MOV AH,2
INT 13H                 ;读入BOOT扇区和FAT,数据存放地
                        ;址仍为800H

```

;计算硬盘数据区的起始逻辑扇区号

```

MOV AX,[BX+0BH]         ;每扇区的字节数
MOV D_020A,AX
MOV AX,[BX+18H]         ;每道扇区数
MOV D_020C,AX
MOV AX,[BX+1AH]         ;磁头数
MOV D_020E,AX
MOV AL,[BX+0DH]         ;每簇扇区数
XOR AH,AH
MOV DS:[0F0H],AX
MOV AX,[BX+16H]         ;一个FAT占用扇区数
MUL BYTE PTR [BX+10H]   ;剩FAT数目=FAT所占总扇区数

```

ADD	AX,[BX+0EH]	;加保留扇区数
ADD	AX,[BX+1CH]	;加隐含扇区数=根目录起始逻辑扇区号
MOV	DS:[0F4H],AX	
MOV	AX,20H	;每个目录项长度=32字节
MUL	WORD PTR [BX+11H]	;乘以根目录项数=根目录长度
DIV	D_020A	;除以每扇区字节数=根目录所占扇区数
ADD	DS:[0F4H],AX	;加根目录起始逻辑扇区号=数据区起始逻辑扇区号

;建立 HZK16 的扇区链表

MOV	BX,0A00H	;FAT首址
MOV	BP,2C2H	;链表首址
CALL	L_0300	;调用建立链表子程序

;初始化常用字区

MOV	AX,BP	;L_0300返回时, BP为可用内存首址
MOV	D_0204,AX	;常用字区首址
MOV	D_0206,AX	;常用字已使用区尾址
MOV	D_0208,AX	;常用字区更新地址,此时常用字区中没有任何常用字
MOV	AL,DS:[82H]	;取命令行参数1
CMP	AL,41H	
JB	L_018F	;是数字(<'A')转
SUB	AL,7	; 'A'-'F' -> 4a-4f

L_018F:

AND	AL,0FH	;30-4f->0-f
XOR	AH,AH	;AXx50=常用字数
MOV	DX,6A4H	;每50个字的缓冲区使用量

;每个汉字本身占2个字节,加上32字节的点阵数据,为34字节. 50x(32+2)=6A4H

MUL	DX	;AX=常用字区长度
-----	----	------------

;初始化内部词组区,内部词组区紧接在常用字区之后

ADD	AX,BP	;AX=内部词组区首址
		;或常用字区尾址
MOV	D_0200,AX	
MOV	AL,2	;缺省内部词组区为2K字节
CMP	BYTE PTR DS:[80H],3	;有命令参数2吗?
JB	L_01AB	;没有,按缺省值设置
MOV	AL,DS:[83H]	;取命令行参数2
AND	AL,0FH	;将数字转换为相应的数值, AL=内部词组区长度(单位:KB)

L_01AB:

XOR	AH,AH	
MOV	DX,400H	;1K=1024字节
MUL	DX	;AX=内部词组长度(单位:字节)
ADD	AX,D_0200	;AX=内部词组区尾址
MOV	D_0202,AX	

;设置 INT 7FH,并驻留退出

INC	AX	;留点余量
PUSH	AX	;保存程序尾址

	MOV	DX,OFFSET L_0211	:INT 7FH的入口地址
	MOV	AX,257FH	
	INT	21H	:设置INT 7FH
	POP	DX	:DX=程序尾址
	INT	27H	:驻留退出
	DB	26 DUP (0)	:没有用
D_01E0	DB	3	:HZK16的文件控制块(FCB)
	DB	'HZK16	:文件名
	DB	20 DUP (0)	
D_0200	DW	0	:内部词组区首址
D_0202	DW	0	:内部词组区尾址
D_0204	DW	0	:常用字区首址
D_0206	DW	0	:常用字已使用区尾址
D_0208	DW	0	:常用字更新地址
D_020A	DW	0	:每扇区字节数
D_020C	DW	0	:每道扇区数
D_020E	DW	0	:磁头数
	DB	0	:没有用
:INT 7FH			
:输入: DX = 汉字			
:输出: DX:0 = 汉字点阵数据地址			
L_0211:	PUSH	DS	
	PUSH	ES	
	PUSH	AX	
	PUSH	BX	
	PUSH	CX	
	PUSH	SI	
	PUSH	DI	:保存寄存器
	PUSH	CS	
	POP	DS	:DS=CS
	PUSH	CS	
	POP	ES	:ES=CS
	MOV	SI,D_0204	:SI=常用字区首址
	JMP	SHORT L_0226	
	NOP		
L_0223:	ADD	SI,20H	:SI=下一个常用字地址
L_0226:	CMP	SI,D_0206	
	JE	L_0242	:常用字区全部查完,没有找到指
			:定汉字,转从磁盘上读入
	LODSW		:取汉字
	CMP	AX,DX	:是否指定汉字吗?
	JNE	L_0223	:不是